

```
! sat20.inp
! 160811 17.0
```

```
! --- sat20
! Basis
```

```
fini
/cle
```

```
/fil,sat20_bas
```

```
! --- Grafik vorbereiten
!/show,JPEG
! reverse video
/RGB,INDEX,100,100,100,0
/RGB,INDEX,80,80,80,13
/RGB,INDEX,60,60,60,14
/RGB,INDEX,0,0,0,15
```

```
!-----
!--- Basismodell
!-----
```

```
! --- Preprocessing
/prep7
```

```
!-----
matpoi1=1
matpoi2=2
/inp,1_0570-20,mat
!-----
```

```
! -----
! Geometrie
et,1,solid45
block,,1,,1,,1
```

```
! --- Vernetzung
esize,,1
mat,1
vmesh,1
```

```
! --- Randbedingungen
! Einspannung an den Seiten
nsel,s,loc,x
nsel,a,loc,x,1
d,all,ux
alls
! ansonsten statisch bestimmt
d,node(0,0,0),uy
d,node(0,0,0),uz
d,node(0,1,0),uz
alls
```

```
! --- Loesungsabschnitt
/solu
nlgeom,on
autots,on
outres,all,all
lnsrch,on
```

```
! Startpunkt
```

```
time,1
nsub,1,1,1
bfe,all,temp,,20
alls
solve
```

```
! Aufheizen
time,2
nsub,50,1000,50
bfe,all,temp,,1500
alls
solve
```

```
! Abkuehlen bis 20
! Material umschalten
mpchg,2,all
time,3
nsub,50,1000,50
bfe,all,temp,,20
alls
solve
```

```
!-----
!--- Basis Ende
!-----
fini
```

```
!-----
!--- Postprocessing
!-----
```

```
!-----
!--- Basis
/fil,sat20_bas
! --- Postprocessing
/post26
file,sat20_bas,rst
esol,2,1,1,bfe,temp
esol,3,1,1,s,x,TRIP 100%
xvar,2
/xrang,0,1500
/yrang,-500.e6,500.e6
/axlab,x,TEMP
/axlab,y,Spannung
/color,curve,blue,1
plvar,3
/color,curve,cyan,1
fini
```

```
! 1_0570.mat
! steel 1.0570 S355J2G3 material data
! 160811 17.0
```

```
! m N s kg W J
```

```
! Basis
```

```
!-----
!--- Aufheizen
```

```

! Als pointer hierfuer ist matpoi1 eingetragen
! -----
! lambda kxx [ W/(m K)]
! -----
! als Funktion der Temperatur
anzahl=6
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 100
tmp(3)= 200
tmp(4)= 400
tmp(5)= 800
tmp(6)=2000
! Funktionswerte
wert(1)= 40.0
wert(2)= 41.5
wert(3)= 42.0
wert(4)= 43.0
wert(5)= 45.0
wert(6)= 50.0
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,kxx,matpoi1,iwert,wert(iwert)
*enddo
! Diagnose
!mplt,kxx,matpoi1
! -----
! rho dens [kg/m**3]
! -----
mp,dens,matpoi1,7.810e3
! -----
! cp c [J/(kg K)]
! -----
! als Funktion der Temperatur
anzahl=4
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 500
tmp(3)=1000
tmp(4)=2000
! Funktionswerte
wert(1)= 500.0
wert(2)= 450.0
wert(3)= 400.0
wert(4)= 400.0
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo

```

```

*do,iwert,1,anzahl
mpdata,c,matpoi1,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,c,matpoi1
! -----
! ex E-Modul [N/m**2]
! -----
! als Funktion der Temperatur
anzahl=7
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 300
tmp(3)= 600
tmp(4)= 800
tmp(5)=1000
tmp(6)=1200
tmp(7)=2000
! Funktionswerte
wert(1)=210000.e6
wert(2)=190000.e6
wert(3)=170000.e6
wert(4)=130000.e6
wert(5)=100000.e6
wert(6)= 50000.e6
wert(7)= 1000.e6
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,ex,matpoi1,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,ex,matpoi1
! -----
! nuxy Querkontraktion [-]
! -----
mp,nuxy,matpoi1,0.3
! -----
! reft Referenz-Temperatur [°C]
! -----
mp,reft,matpoi1,20
! -----
! alpx thermischer Ausdehnungskoeffizient
! indirekt eingegeben ueber
! thsx thermische Dehnungen [1/°C]
! -----
! als Funktion der Temperatur
anzahl=9
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 0
tmp(2)= 200

```

```

tmp(3)= 400
tmp(4)= 600
tmp(5)= 800
tmp(6)= 900
tmp(7)=1200
tmp(8)=1500
tmp(9)=2000
! Funktionswerte
wert(1)= 0.0
wert(2)= 2.40e-3
wert(3)= 4.80e-3
wert(4)= 7.20e-3
wert(5)= 9.6e-3
wert(6)= 8.5e-3
wert(7)= 13.3e-3
wert(8)= 18.1e-3
wert(9)= 24.5e-3
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,thsx,matpoi1,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,thsx,matpoi1
! -----
! Plastizitaet Temp-abhaengig
! -----
! als Funktion der Temperatur
! elastische Grenze [N/m**2], Tangentenmodul [N/m**2]
anzahl=6
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
wert2=
*dim,wert2,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 350
tmp(3)= 600
tmp(4)= 900
tmp(5)=1000
tmp(6)=1500
! Funktionswert 1: elastische Grenze [N/m**2]
wert(1)= 350.0e6
wert(2)= 240.0e6
wert(3)= 150.0e6
wert(4)= 140.0e6
wert(5)= 121.0e6
wert(6)= 30.0e6
! Funktionswert 2: Tangentenmodul [N/m**2]
wert2(1)= 20000.0e6
wert2(2)= 17900.0e6
wert2(3)= 15000.0e6
wert2(4)= 9817.0e6
wert2(5)= 8348.0e6
wert2(6)= 1000.0e6
! uebertragen
tbdele,bkin,matpoi1

```

```

tb,bkin,matpoi1,anzahl
*do,iwert,1,anzahl
tbtemp,tmp(iwert)
tbdata,,wert(iwert),wert2(iwert)
*enddo
! Diagnose
!tbpl,bkin,matpoi1
! -----

!-----
!--- Abkuehlen
! Als pointer hierfuer ist matpoi2 eingetragen
! TRIP im Umwandlungsbereich
! -----
! lambda kxx [ W/(m K)]
! -----
! als Funktion der Temperatur
anzahl=6
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 100
tmp(3)= 200
tmp(4)= 400
tmp(5)= 800
tmp(6)=2000
! Funktionswerte
wert(1)= 40.0
wert(2)= 41.5
wert(3)= 42.0
wert(4)= 43.0
wert(5)= 45.0
wert(6)= 50.0
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,kxx,matpoi2,iwert,wert(iwert)
*enddo
! Diagnose
!mplt,kxx,matpoi2
! -----
! rho dens [kg/m**3]
! -----
mp,dens,matpoi2,7.810e3
! -----
! cp c [J/(kg K)]
! -----
! als Funktion der Temperatur
anzahl=4
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 500

```

```

tmp(3)=1000
tmp(4)=2000
! Funktionswerte
wert(1)= 500.0
wert(2)= 450.0
wert(3)= 400.0
wert(4)= 400.0
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,c,matpoi2,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,c,matpoi2
! -----
! ex E-Modul [N/m**2]
! -----
! als Funktion der Temperatur
anzahl=7
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 300
tmp(3)= 600
tmp(4)= 800
tmp(5)=1000
tmp(6)=1200
tmp(7)=2000
! Funktionswerte
wert(1)=210000.e6
wert(2)=190000.e6
wert(3)=170000.e6
wert(4)=130000.e6
wert(5)=100000.e6
wert(6)= 50000.e6
wert(7)= 1000.e6
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,ex,matpoi2,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,ex,matpoi2
! -----
! nuxy Querkontraktion [-]
! -----
mp,nuxy,matpoi2,0.3
! -----
! reft Referenz-Temperatur [°C]
! -----
mp,reft,matpoi2,20
! -----
! alp thermischer Ausdehnungskoeffizient

```

```

! indirekt eingegeben ueber
! thsx thermische Dehnungen [1/°C]
! -----
! als Funktion der Temperatur
anzahl=9
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
! Temperaturen
tmp(1)= 0
tmp(2)= 200
tmp(3)= 300
tmp(4)= 400
tmp(5)= 600
tmp(6)= 900
tmp(7)=1200
tmp(8)=1500
tmp(9)=2000
! Funktionswerte
wert(1)= 0.0
wert(2)= 2.4e-3
wert(3)= -1.10e-3
wert(4)= 0.5e-3
wert(5)= 3.7e-3
wert(6)= 8.5e-3
wert(7)= 13.3e-3
wert(8)= 18.1e-3
wert(9)= 24.5e-3
! uebertragen
mptemp
*do,iwert,1,anzahl
mptemp,iwert,tmp(iwert)
*enddo
*do,iwert,1,anzahl
mpdata,thsx,matpoi2,iwert,wert(iwert)
*enddo
! Diagnose
!mppl,thsx,matpoi2
! -----
! Plastizitaet Temp-abhaengig
! -----
! als Funktion der Temperatur
! elastische Grenze [N/m**2], Tangentenmodul [N/m**2]
anzahl=6
tmp=
*dim,tmp,,anzahl
wert=
*dim,wert,,anzahl
wert2=
*dim,wert2,,anzahl
! Temperaturen
tmp(1)= 20
tmp(2)= 180
tmp(3)= 200
tmp(4)= 300
tmp(5)= 320
tmp(6)=1500
! Funktionswert 1: elastische Grenze [N/m**2]
wert(1)= 350.0e6
wert(2)= 290.0e6
wert(3)= 50.0e6

```



```
wert(4)= 50.0e6
wert(5)= 240.0e6
wert(6)= 30.0e6
! Funktionswert 2: Tangentenmodul [N/m**2]
wert2(1)= 20000.0e6
wert2(2)= 18855.0e6
wert2(3)= 10000.0e6
wert2(4)= 10000.0e6
wert2(5)= 17900.0e6
wert2(6)= 1000.0e6
! uebertragen
tbdele,bkin,matpoi2
tb,bkin,matpoi2,anzahl
*do,iwert,1,anzahl
tbtemp,tmp(iwert)
tbdata,,wert(iwert),wert2(iwert)
*enddo
! Diagnose
!tbpl,bkin,matpoi2
! -----
```